Rules for integrands of the form $u\left(a+b x+c x^2+d x^3+e x^4\right)^p$

1. $\int u\left(a+b x+c x^2+b x^3+a x^4\right)^p \, dx$ when $a==e \land b==d$

   1. $\int \frac{f+g x^2}{\left(d+e x+d x^2\right)\sqrt{a+b x+c x^2+b x^3+a x^4}} \, dx$ when $b d-a e==0 \land f+g==0$

      1: $\int \frac{f+g x^2}{\left(d+e x+d x^2\right)\sqrt{a+b x+c x^2+b x^3+a x^4}} \, dx$ when $b d-a e==0 \land f+g==0 \land a^2 (2 a-c)>0$

- **Rule: If** $b d-a e==0 \land f+g==0 \land a^2 (2 a-c)>0$, **then**

$$\int \frac{f+g x^2}{\left(d+e x+d x^2\right)\sqrt{a+b x+c x^2+b x^3+a x^4}} \, dx \longrightarrow \frac{a f}{d\sqrt{a^2 (2 a-c)}} \text{ArcTan}\left[\frac{a b+\left(4 a^2+b^2-2 a c\right) x+a b x^2}{2\sqrt{a^2 (2 a-c)}\sqrt{a+b x+c x^2+b x^3+a x^4}}\right]$$

- **Program code:**

```
Int[(f_+g_.*x_^2)/((d_+e_.*x_+d_.*x_^2)*Sqrt[a_+b_.*x_+c_.*x_^2+b_.*x_^3+a_.*x_^4]),x_Symbol] :=
  a*f/(d*Rt[a^2*(2*a-c),2])*ArcTan[(a*b+(4*a^2+b^2-2*a*c)*x+a*b*x^2)/(2*Rt[a^2*(2*a-c),2]*Sqrt[a+b*x+c*x^2+b*x^3+a*x^4])] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*d-a*e,0] && EqQ[f+g,0] && PosQ[a^2*(2*a-c)]
```

      2: $\int \frac{f+g x^2}{\left(d+e x+d x^2\right)\sqrt{a+b x+c x^2+b x^3+a x^4}} \, dx$ when $b d-a e==0 \land f+g==0 \land a^2 (2 a-c)\not>0$

- **Rule: If** $b d-a e==0 \land f+g==0 \land a^2 (2 a-c)\not>0$, **then**

$$\int \frac{f+g x^2}{\left(d+e x+d x^2\right)\sqrt{a+b x+c x^2+b x^3+a x^4}} \, dx \longrightarrow -\frac{a f}{d\sqrt{-a^2 (2 a-c)}} \text{ArcTanh}\left[\frac{a b+\left(4 a^2+b^2-2 a c\right) x+a b x^2}{2\sqrt{-a^2 (2 a-c)}\sqrt{a+b x+c x^2+b x^3+a x^4}}\right]$$

- **Program code:**

```
Int[(f_+g_.*x_^2)/((d_+e_.*x_+d_.*x_^2)*Sqrt[a_+b_.*x_+c_.*x_^2+b_.*x_^3+a_.*x_^4]),x_Symbol] :=
  -a*f/(d*Rt[-a^2*(2*a-c),2])*ArcTanh[(a*b+(4*a^2+b^2-2*a*c)*x+a*b*x^2)/(2*Rt[-a^2*(2*a-c),2]*Sqrt[a+b*x+c*x^2+b*x^3+a*x^4])] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*d-a*e,0] && EqQ[f+g,0] && NegQ[a^2*(2*a-c)]
```

   2. $\int \frac{u\left(A+B x+C x^2+D x^3\right)}{a+b x+c x^2+b x^3+a x^4} \, dx$

1: $\int \frac{A+B x+C x^2+D x^3}{a+b x+c x^2+b x^3+a x^4} \, dx$ when $8 a^2+b^2-4 a c>0$

- **Derivation: Algebraic expansion**

- **Basis: Let** $q \to \sqrt{8 a^2+b^2-4 a c}$**, then** $\frac{A+B x+C x^2+D x^3}{a+b x+c x^2+b x^3+a x^4} == \frac{b A-2 a B+2 a D+A q+(2 a A-2 a C+b D+D q) x}{q\left(2 a+(b+q) x+2 a x^2\right)}-\frac{b A-2 a B+2 a D-A q+(2 a A-2 a C+b D-D q) x}{q\left(2 a+(b-q) x+2 a x^2\right)}$

- **Rule: If** $8 a^2+b^2-4 a c>0$**, let** $q \to \sqrt{8 a^2+b^2-4 a c}$**, then**

$$\int \frac{A+B x+C x^2+D x^3}{a+b x+c x^2+b x^3+a x^4} \, dx \longrightarrow$$
$$\frac{1}{q} \int \frac{b A-2 a B+2 a D+A q+(2 a A-2 a C+b D+D q) x}{2 a+(b+q) x+2 a x^2} \, dx-\frac{1}{q} \int \frac{b A-2 a B+2 a D-A q+(2 a A-2 a C+b D-D q) x}{2 a+(b-q) x+2 a x^2} \, dx$$

- **Program code:**

```
Int[P3_/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Sqrt[8*a^2+b^2-4*a*c],A=Coeff[P3,x,0],B=Coeff[P3,x,1],C=Coeff[P3,x,2],D=Coeff[P3,x,3]},
  1/q*Int[(b*A-2*a*B+2*a*D+A*q+(2*a*A-2*a*C+b*D+D*q)*x)/(2*a+(b+q)*x+2*a*x^2),x] -
  1/q*Int[(b*A-2*a*B+2*a*D-A*q+(2*a*A-2*a*C+b*D-D*q)*x)/(2*a+(b-q)*x+2*a*x^2),x]] /;
FreeQ[{a,b,c},x] && PolyQ[P3,x,3] && EqQ[a,e] && EqQ[b,d] && SumQ[Factor[a+b*x+c*x^2+b*x^3+a*x^4]]
```

2: $\int \frac{x^m \left(A + B x + C x^2 + D x^3\right)}{a + b x + c x^2 + b x^3 + a x^4} \, dx$ when $8 a^2 + b^2 - 4 a c > 0$

- **Derivation: Algebraic expansion**

- **Basis: Let** $q \to \sqrt{8 a^2 + b^2 - 4 a c}$**, then** $\frac{A+B x+C x^2+D x^3}{a+b x+c x^2+b x^3+a x^4} == \frac{b A-2 a B+2 a D+A q+(2 a A-2 a C+b D+D q) x}{q (2 a+(b+q) x+2 a x^2)} - \frac{b A-2 a B+2 a D-A q+(2 a A-2 a C+b D-D q) x}{q (2 a+(b-q) x+2 a x^2)}$

- **Rule: If** $8 a^2 + b^2 - 4 a c > 0$**, let** $q \to \sqrt{8 a^2 + b^2 - 4 a c}$**, then**

$$\int \frac{x^m \left(A + B x + C x^2 + D x^3\right)}{a + b x + c x^2 + b x^3 + a x^4} \, dx \longrightarrow$$
$$\frac{1}{q} \int \frac{x^m \left(b A - 2 a B + 2 a D + A q + (2 a A - 2 a C + b D + D q) x\right)}{2 a + (b + q) x + 2 a x^2} \, dx - \frac{1}{q} \int \frac{x^m \left(b A - 2 a B + 2 a D - A q + (2 a A - 2 a C + b D - D q) x\right)}{2 a + (b - q) x + 2 a x^2} \, dx$$

- **Program code:**

```
Int[x_^m_.*P3_/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Sqrt[8*a^2+b^2-4*a*c],A=Coeff[P3,x,0],B=Coeff[P3,x,1],C=Coeff[P3,x,2],D=Coeff[P3,x,3]},
  1/q*Int[x^m*(b*A-2*a*B+2*a*D+A*q+(2*a*A-2*a*C+b*D+D*q)*x)/(2*a+(b+q)*x+2*a*x^2),x] -
  1/q*Int[x^m*(b*A-2*a*B+2*a*D-A*q+(2*a*A-2*a*C+b*D-D*q)*x)/(2*a+(b-q)*x+2*a*x^2),x]] /;
FreeQ[{a,b,c,m},x] && PolyQ[P3,x,3] && EqQ[a,e] && EqQ[b,d] && SumQ[Factor[a+b*x+c*x^2+b*x^3+a*x^4]]
```

2. $\int u \left(a + b x + c x^2 + d x^3 + e x^4\right)^p dx$ when $d^3 - 4 c d e + 8 b e^2 == 0$

1: $\int \left(a + b x + c x^2 + d x^3 + e x^4\right)^p dx$ when $d^3 - 4 c d e + 8 b e^2 == 0 \land p \notin \{1, 2, 3\}$

- **Derivation: Integration by substitution**
- **Basis: If** $d^3 - 4 c d e + 8 b e^2 == 0$**, then** $\left(a + b x + c x^2 + d x^3 + e x^4\right)^p == \text{Subst}\left[\left(a + \frac{d^4}{256 e^3} - \frac{b d}{8 e} + \left(c - \frac{3 d^2}{8 e}\right) x^2 + e x^4\right)^p, x, \frac{d}{4 e} + x\right] \partial_x \left(\frac{d}{4 e} + x\right)$
- **Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial.**
- **Rule: If** $d^3 - 4 c d e + 8 b e^2 == 0 \land p \notin \{1, 2, 3\}$**, then**

$$\int \left(a + b x + c x^2 + d x^3 + e x^4\right)^p dx \longrightarrow \text{Subst}\left[\int \left(a + \frac{d^4}{256 e^3} - \frac{b d}{8 e} + \left(c - \frac{3 d^2}{8 e}\right) x^2 + e x^4\right)^p dx, x, \frac{d}{4 e} + x\right]$$

- **Program code:**

```
Int[P4_^p_,x_Symbol] :=
  With[{a=Coeff[P4,x,0],b=Coeff[P4,x,1],c=Coeff[P4,x,2],d=Coeff[P4,x,3],e=Coeff[P4,x,4]},
  Subst[Int[SimplifyIntegrand[(a+d^4/(256*e^3)-b*d/(8*e)+(c-3*d^2/(8*e))*x^2+e*x^4)^p,x],x],x,d/(4*e)+x] /;
 EqQ[d^3-4*c*d*e+8*b*e^2,0] && NeQ[d,0]] /;
FreeQ[p,x] && PolyQ[P4,x,4] && NeQ[p,2] && NeQ[p,3]
```

2: $\int P_q[x] \left(a+b x+c x^2+d x^3+e x^4\right)^p dx$ when $d^3-4 c d e+8 b e^2==0 \land p \notin \mathbb{Z}^+$

- **Derivation: Integration by substitution**
- **Basis: If** $d^3-4 c d e+8 b e^2==0$**, then**
$F[x] \left(a+b x+c x^2+d x^3+e x^4\right)^p == \text{Subst}\left[F\left[-\frac{d}{4 e}+x\right] \left(a+\frac{d^4}{256 e^3}-\frac{b d}{8 e}+\left(c-\frac{3 d^2}{8 e}\right) x^2+e x^4\right)^p,\ x,\ \frac{d}{4 e}+x\right] \partial_x \left(\frac{d}{4 e}+x\right)$
- **Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial.**
- **Rule: If** $d^3-4 c d e+8 b e^2==0 \land p \notin \mathbb{Z}^+$**, then**

$$\int P_q[x] \left(a+b x+c x^2+d x^3+e x^4\right)^p dx \longrightarrow \text{Subst}\left[\int P_q\left[t-\frac{d}{4 e}\right] \left(a+\frac{d^4}{256 e^3}-\frac{b d}{8 e}+\left(c-\frac{3 d^2}{8 e}\right) x^2+e x^4\right)^p dx,\ x,\ \frac{d}{4 e}+x\right]$$

- **Program code:**

```
Int[Pq_*Q4_^p_,x_Symbol] :=
  With[{a=Coeff[Q4,x,0],b=Coeff[Q4,x,1],c=Coeff[Q4,x,2],d=Coeff[Q4,x,3],e=Coeff[Q4,x,4]},
  Subst[Int[SimplifyIntegrand[ReplaceAll[Pq,x→-d/(4*e)+x]*(a+d^4/(256*e^3)-b*d/(8*e)+(c-3*d^2/(8*e))*x^2+e*x^4)^p,x],x],x,d/(4*e)+x
 EqQ[d^3-4*c*d*e+8*b*e^2,0] && NeQ[d,0]] /;
FreeQ[p,x] && PolyQ[Pq,x] && PolyQ[Q4,x,4] && Not[IGtQ[p,0]]
```

3. $\int u\,(a+b\,x+c\,x^2+d\,x^3+e\,x^4)^p\,dx$ when $b^3-4\,a\,b\,c+8\,a^2\,d=0$

1: $\int (a+b\,x+c\,x^2+d\,x^3+e\,x^4)^p\,dx$ when $b^3-4\,a\,b\,c+8\,a^2\,d=0\land 2\,p\in\mathbb{Z}$

- **Derivation: Integration by substitution**
- **Basis: If** $b^3-4\,a\,b\,c+8\,a^2\,d=0$**, then**
$(a+b\,x+c\,x^2+d\,x^3+e\,x^4)^p=-16\,a^2\,\text{Subst}\left[\frac{1}{(b-4\,a\,x)^2}\left(\frac{a\,(-3\,b^4+16\,a\,b^2\,c-64\,a^2\,b\,d+256\,a^3\,e-32\,a^2\,(3\,b^2-8\,a\,c)\,x^2+256\,a^4\,x^4)}{(b-4\,a\,x)^4}\right)^p,\ x,\ \frac{b}{4\,a}+\frac{1}{x}\right]\partial_x\left(\frac{b}{4\,a}+\frac{1}{x}\right)$
- **Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial over the 4th power of a linear.**
- **Rule: If** $b^3-4\,a\,b\,c+8\,a^2\,d=0\land 2\,p\in\mathbb{Z}$**, then**

$$\int (a+b\,x+c\,x^2+d\,x^3+e\,x^4)^p\,dx\ \longrightarrow$$

$$-16\,a^2\,\text{Subst}\left[\int\frac{1}{(b-4\,a\,x)^2}\left(\frac{1}{(b-4\,a\,x)^4}a\,\left(-3\,b^4+16\,a\,b^2\,c-64\,a^2\,b\,d+256\,a^3\,e-32\,a^2\,\left(3\,b^2-8\,a\,c\right)\,x^2+256\,a^4\,x^4\right)\right)^p dx,\ x,\ \frac{b}{4\,a}+\frac{1}{x}\right]$$

- **Program code:**

```
Int[P4_^p_,x_Symbol] :=
  With[{a=Coeff[P4,x,0],b=Coeff[P4,x,1],c=Coeff[P4,x,2],d=Coeff[P4,x,3],e=Coeff[P4,x,4]},
  -16*a^2*Subst[
    Int[1/(b-4*a*x)^2*(a*(-3*b^4+16*a*b^2*c-64*a^2*b*d+256*a^3*e-32*a^2*(3*b^2-8*a*c)*x^2+256*a^4*x^4)/(b-4*a*x)^4)^p,x],
    x,b/(4*a)+1/x] /;
 NeQ[a,0] && NeQ[b,0] && EqQ[b^3-4*a*b*c+8*a^2*d,0]] /;
FreeQ[p,x] && PolyQ[P4,x,4] && IntegerQ[2*p]
```

4. $\int \frac{u}{a+b\,x+c\,x^2+d\,x^3+e\,x^4}\,dx$

1. $\int \frac{A+B\,x+C\,x^2}{a+b\,x+c\,x^2+d\,x^3+e\,x^4}\,dx$ when $B^2\,d+2\,C\,(b\,C+A\,d)-2\,B\,(c\,C+2\,A\,e)=0\land 2\,B^2\,c\,C-8\,a\,C^3-B^3\,d-4\,A\,B\,C\,d+4\,A\,\left(B^2+2\,A\,C\right)\,e=0$

1: $\int \frac{A+B\,x+C\,x^2}{a+b\,x+c\,x^2+d\,x^3+e\,x^4}\,dx$ when
$B^2\,d+2\,C\,(b\,C+A\,d)-2\,B\,(c\,C+2\,A\,e)=0\land 2\,B^2\,c\,C-8\,a\,C^3-B^3\,d-4\,A\,B\,C\,d+4\,A\,\left(B^2+2\,A\,C\right)\,e=0\land C\,\left(2\,e\,(B\,d-4\,A\,e)+C\,\left(d^2-4\,c\,e\right)\right)>0$

- **Rule: If** $B^2\,d+2\,C\,(b\,C+A\,d)-2\,B\,(c\,C+2\,A\,e)=0\ \land$
$2\,B^2\,c\,C-8\,a\,C^3-B^3\,d-4\,A\,B\,C\,d+4\,A\,\left(B^2+2\,A\,C\right)\,e=0\land C\,\left(2\,e\,(B\,d-4\,A\,e)+C\,\left(d^2-4\,c\,e\right)\right)>0$,
let $q\to\sqrt{C\,\left(2\,e\,(B\,d-4\,A\,e)+C\,\left(d^2-4\,c\,e\right)\right)}$**, then**

$$\int \frac{A + B x + C x^2}{a + b x + c x^2 + d x^3 + e x^4} \, dx \longrightarrow$$

$$-\frac{2 C^2}{q} \text{ArcTanh}\left[\frac{C d - B e + 2 C e x}{q}\right] +$$

$$\frac{2 C^2}{q} \text{ArcTanh}\left[\frac{1}{q \left(B^2 - 4 A C\right)} C \left(4 B c C - 3 B^2 d - 4 A C d + 12 A B e + 4 C (2 c C - B d + 2 A e) x + 4 C (2 C d - B e) x^2 + 8 C^2 e x^3\right)\right]$$

- **Program code:**

```
Int[(A_.+B_.*x_+C_.*x_^2)/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Rt[C*(2*e*(B*d-4*A*e)+C*(d^2-4*c*e)),2]},
  -2*C^2/q*ArcTanh[(C*d-B*e+2*C*e*x)/q] +
  2*C^2/q*ArcTanh[C*(4*B*c*C-3*B^2*d-4*A*C*d+12*A*B*e+4*C*(2*c*C-B*d+2*A*e)*x+4*C*(2*C*d-B*e)*x^2+8*C^2*e*x^3)/(q*(B^2-4*A*C))]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && EqQ[B^2*d+2*C*(b*C+A*d)-2*B*(c*C+2*A*e),0] &&
  EqQ[2*B^2*c*C-8*a*C^3-B^3*d-4*A*B*C*d+4*A*(B^2+2*A*C)*e,0] && PosQ[C*(2*e*(B*d-4*A*e)+C*(d^2-4*c*e))]
```

```
Int[(A_.+C_.*x_^2)/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Rt[C*(-8*A*e^2+C*(d^2-4*c*e)),2]},
  -2*C^2/q*ArcTanh[C*(d+2*e*x)/q] + 2*C^2/q*ArcTanh[C*(A*d-2*(c*C+A*e)*x-2*C*d*x^2-2*C*e*x^3)/(A*q)]] /;
FreeQ[{a,b,c,d,e,A,C},x] && EqQ[b*C+A*d,0] && EqQ[a*C^2-A^2*e,0] && PosQ[C*(-8*A*e^2+C*(d^2-4*c*e))]
```

2: $\int \frac{A+B x+C x^2}{a+b x+c x^2+d x^3+e x^4} dx$ when

$B^2 d+2 C (b C+A d)-2 B (c C+2 A e)==0 \land 2 B^2 c C-8 a C^3-B^3 d-4 A B C d+4 A \left(B^2+2 A C\right) e==0 \land C \left(2 e (B d-4 A e)+C \left(d^2-4 c e\right)\right) \not> 0$

- **Rule: If** $B^2 d+2 C (b C+A d)-2 B (c C+2 A e)==0 \land$,
  $2 B^2 c C-8 a C^3-B^3 d-4 A B C d+4 A \left(B^2+2 A C\right) e==0 \land C \left(2 e (B d-4 A e)+C \left(d^2-4 c e\right)\right) \not> 0$
  let $q=\sqrt{-C \left(2 e (B d-4 A e)+C \left(d^2-4 c e\right)\right)}$ **, then**

$$\int \frac{A+B x+C x^2}{a+b x+c x^2+d x^3+e x^4} dx \longrightarrow$$

$$\frac{2 C^2}{q} \text{ArcTan}\left[\frac{C d-B e+2 C e x}{q}\right]-\frac{2 C^2}{q} \text{ArcTan}\left[\frac{1}{q \left(B^2-4 A C\right)} C \left(4 B c C-3 B^2 d-4 A C d+12 A B e+4 C (2 c C-B d+2 A e) x+4 C (2 C d-B e) x^2+8 C^2 e x^3\right)\right]$$

- **Program code:**

```
Int[(A_.+B_.*x_+C_.*x_^2)/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Rt[-C*(2*e*(B*d-4*A*e)+C*(d^2-4*c*e)),2]},
  2*C^2/q*ArcTan[(C*d-B*e+2*C*e*x)/q] -
  2*C^2/q*ArcTan[C*(4*B*c*C-3*B^2*d-4*A*C*d+12*A*B*e+4*C*(2*c*C-B*d+2*A*e)*x+4*C*(2*C*d-B*e)*x^2+8*C^2*e*x^3)/(q*(B^2-4*A*C))]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && EqQ[B^2*d+2*C*(b*C+A*d)-2*B*(c*C+2*A*e),0] &&
  EqQ[2*B^2*c*C-8*a*C^3-B^3*d-4*A*B*C*d+4*A*(B^2+2*A*C)*e,0] && NegQ[C*(2*e*(B*d-4*A*e)+C*(d^2-4*c*e))]
```

```
Int[(A_.+C_.*x_^2)/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{q=Rt[-C*(-8*A*e^2+C*(d^2-4*c*e)),2]},
  2*C^2/q*ArcTan[(C*d+2*C*e*x)/q] - 2*C^2/q*ArcTan[-C*(-A*d+2*(c*C+A*e)*x+2*C*d*x^2+2*C*e*x^3)/(A*q)]] /;
FreeQ[{a,b,c,d,e,A,C},x] && EqQ[b*C+A*d,0] && EqQ[a*C^2-A^2*e,0] && NegQ[C*(-8*A*e^2+C*(d^2-4*c*e))]
```

2: $\int \frac{A + B x + C x^2 + D x^3}{a + b x + c x^2 + d x^3 + e x^4} \, dx$ when $4 d (c D - 2 B e)^2 + 8 (3 d D - 4 C e) (b d D - b C e - A d e) - 4 (c D - 2 B e) \left(3 c d D - 4 c C e + 2 b D e - 8 A e^2\right) == 0$ and $8 d (c D - 2 B e)^3 + 8 d (b D - 4 A e) (c D - 2 B e) (3 d D - 4 C e) + 8 a (3 d D - 4 C e)^3 - 8 c (c D - 2 B e)^2 (3 d D - 4 C e) - 4 e (b D - 4 A e) \left(4 (c D - 2 B e)^2 + 2 (b D - 4 A e) (3 d D - 4 C e)\right) == 0$

- **Derivation: Algebraic expansion**
- **Basis:** $A + B x + C x^2 + D x^3 == \frac{D}{4 e} \left(b + 2 c x + 3 d x^2 + 4 e x^3\right) - \frac{1}{4 e} \left(b D - 4 A e + 2 (c D - 2 B e) x + (3 d D - 4 C e) x^2\right)$
- **Note: Resulting integrand is of the form** $\frac{A+B x+C x^2}{a+b x+c x^2+d x^3+e x^4}$ **where** $B^2 d + 2 C (b C + A d) - 2 B (c C + 2 A e) == 0 \wedge 2 B^2 c C - 8 a C^3 - B^3 d - 4 A B C d + 4 A \left(B^2 + 2 A C\right) e == 0$.
- **Rule: If** $4 d (c D - 2 B e)^2 + 8 (3 d D - 4 C e) (b d D - b C e - A d e) - 4 (c D - 2 B e) \left(3 c d D - 4 c C e + 2 b D e - 8 A e^2\right) == 0$ **and**
- $8 d (c D - 2 B e)^3 + 8 d (b D - 4 A e) (c D - 2 B e) (3 d D - 4 C e) + 8 a (3 d D - 4 C e)^3 - 8 c (c D - 2 B e)^2 (3 d D - 4 C e) - 4 e (b D - 4 A e) \left(4 (c D - 2 B e)^2 + 2 (b D - 4 A e) (3 d D - 4 C e)\right) == 0$ **, then**

$$\int \frac{A + B x + C x^2 + D x^3}{a + b x + c x^2 + d x^3 + e x^4} \, dx \to \frac{D}{4 e} \int \frac{b + 2 c x + 3 d x^2 + 4 e x^3}{a + b x + c x^2 + d x^3 + e x^4} \, dx - \frac{1}{4 e} \int \frac{b D - 4 A e + 2 (c D - 2 B e) x + (3 d D - 4 C e) x^2}{a + b x + c x^2 + d x^3 + e x^4} \, dx$$

$$\to \frac{D}{4 e} \operatorname{Log}\left[a + b x + c x^2 + d x^3 + e x^4\right] - \frac{1}{4 e} \int \frac{b D - 4 A e + 2 (c D - 2 B e) x + (3 d D - 4 C e) x^2}{a + b x + c x^2 + d x^3 + e x^4} \, dx$$

- **Program code:**

```
Int[P3_/(a_+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4),x_Symbol] :=
  With[{A=Coeff[P3,x,0],B=Coeff[P3,x,1],C=Coeff[P3,x,2],D=Coeff[P3,x,3]},
    D/(4*e)*Log[a+b*x+c*x^2+d*x^3+e*x^4] -
    1/(4*e)*Int[(b*D-4*A*e+2*(c*D-2*B*e)*x+(3*d*D-4*C*e)*x^2)/(a+b*x+c*x^2+d*x^3+e*x^4),x] /;
   EqQ[4*d*(c*D-2*B*e)^2+8*(3*d*D-4*C*e)*(b*d*D-b*C*e-A*d*e)-4*(c*D-2*B*e)*(3*c*d*D-4*c*C*e+2*b*D*e-8*A*e^2),0] &&
   EqQ[8*d*(c*D-2*B*e)^3+8*d*(b*D-4*A*e)*(c*D-2*B*e)*(3*d*D-4*C*e)+8*a*(3*d*D-4*C*e)^3-8*c*(c*D-2*B*e)^2*(3*d*D-4*C*e)-
     4*e*(b*D-4*A*e)*(4*(c*D-2*B*e)^2+2*(b*D-4*A*e)*(3*d*D-4*C*e)),0]] /;
FreeQ[{a,b,c,d,e},x] && PolyQ[P3,x,3]
```